# $Id: PKGBUILD 83877 2013-02-06 10:39:22Z spupykin $
# Contributor: Thomas S Hatch <thatch45@gmail.com>
# Contributor: Jon Nordby <jononor@gmail.com>

pkgname=augeas
pkgver=1.0.0
pkgrel=1
pkgdesc="A configuration editing tool that parses config files and transforms them into a tree"
arch=('i686' 'x86_64')
url="http://augeas.net"
license=('LGPL')
depends=('libxml2' 'gcc-libs')
options=('!libtool')
source=(http://augeas.net/download/$pkgname-$pkgver.tar.gz{,.sig})
md5sums=('82131019432ecf8102e1491610ad2dd1'
         '63b2f334b030c957c98ee01ef207cfdd')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  sed -i 's|Requires:.*|Requires: libxml-2.0|' augeas.pc.in
  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
